

NAG Fortran Library Routine Document

E02BDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

E02BDF computes the definite integral of a cubic spline from its B-spline representation.

2 Specification

```
SUBROUTINE E02BDF(NCAP7, LAMDA, C, DEFINT, IFAIL)
INTEGER          NCAP7, IFAIL
real           LAMDA(NCAP7), C(NCAP7), DEFINT
```

3 Description

This routine computes the definite integral of the cubic spline $s(x)$ between the limits $x = a$ and $x = b$, where a and b are respectively the lower and upper limits of the range over which $s(x)$ is defined. It is assumed that $s(x)$ is represented in terms of its B-spline coefficients c_i , for $i = 1, 2, \dots, \bar{n} + 3$ and (augmented) ordered knot set λ_i , for $i = 1, 2, \dots, \bar{n} + 7$, with $\lambda_i = a$, for $i = 1, 2, 3, 4$ and $\lambda_i = b$, for $i = \bar{n} + 4, \bar{n} + 5, \bar{n} + 6, \bar{n} + 7$, (see E02BAF), i.e.,

$$s(x) = \sum_{i=1}^q c_i N_i(x).$$

Here $q = \bar{n} + 3$, \bar{n} is the number of intervals of the spline and $N_i(x)$ denotes the normalised B-spline of degree 3 (order 4) defined upon the knots $\lambda_i, \lambda_{i+1}, \dots, \lambda_{i+4}$.

The method employed uses the formula given in Section 3 of Cox (1975a).

E02BDF can be used to determine the definite integrals of cubic spline fits and interpolants produced by E02BAF.

4 References

Cox M G (1975a) An algorithm for spline interpolation *J. Inst. Math. Appl.* **15** 95–108

5 Parameters

1: NCAP7 – INTEGER *Input*

On entry: $\bar{n} + 7$, where \bar{n} is the number of intervals of the spline (which is one greater than the number of interior knots, i.e., the knots strictly within the range a to b) over which the spline is defined.

Constraint: $\text{NCAP7} \geq 8$.

2: LAMDA(NCAP7) – **real** array *Input*

On entry: LAMDA(j) must be set to the value of the j th member of the complete set of knots, λ_j for $j = 1, 2, \dots, \bar{n} + 7$.

Constraint: the LAMDA(j) must be in non-decreasing order with $\text{LAMDA}(\text{NCAP7} - 3) > \text{LAMDA}(4)$ and satisfy $\text{LAMDA}(1) = \text{LAMDA}(2) = \text{LAMDA}(3) = \text{LAMDA}(4)$ and $\text{LAMDA}(\text{NCAP7} - 3) = \text{LAMDA}(\text{NCAP7} - 2) = \text{LAMDA}(\text{NCAP7} - 1) = \text{LAMDA}(\text{NCAP7})$.

- 3: C(NCAP7) – *real* array *Input*
On entry: the coefficient c_i of the B-spline $N_i(x)$, for $i = 1, 2, \dots, \bar{n} + 3$. The remaining elements of the array are not used.
- 4: DEFINT – *real* *Output*
On exit: the value of the definite integral of $s(x)$ between the limits $x = a$ and $x = b$, where $a = \lambda_4$ and $b = \lambda_{\bar{n}+4}$.
- 5: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

NCAP7 < 8, i.e., the number of intervals is not positive.

IFAIL = 2

At least one of the following restrictions on the knots is violated:

$$\text{LAMDA}(\text{NCAP7} - 3) > \text{LAMDA}(4) ,$$

$$\text{LAMDA}(j) \geq \text{LAMDA}(j - 1),$$

for $j = 2, 3, \dots, \text{NCAP7}$, with equality in the cases $j = 2, 3, 4, \text{NCAP7} - 2, \text{NCAP7} - 1$, and NCAP7.

7 Accuracy

The rounding errors are such that the computed value of the integral is exact for a slightly perturbed set of B-spline coefficients c_i differing in a relative sense from those supplied by no more than $2.2 \times (\bar{n} + 3) \times \text{machine precision}$.

8 Further Comments

The time taken by the routine is approximately proportional to $\bar{n} + 7$.

9 Example

Determine the definite integral over the interval $0 \leq x \leq 6$ of a cubic spline having 6 interior knots at the positions $\lambda = 1, 3, 3, 3, 4, 4$, the 8 additional knots 0, 0, 0, 0, 6, 6, 6, 6, and the 10 B-spline coefficients 10, 12, 13, 15, 22, 26, 24, 18, 14, 12.

The input data items (using the notation of Section 5 of the documents for E02BDF) comprise the following values in the order indicated:

$$\begin{aligned} & \bar{n} \\ & \text{LAMDA}(j), \quad \text{for } j = 1, 2, \dots, \text{NCAP7} \\ & \text{C}(j), \quad \text{for } j = 1, 2, \dots, \text{NCAP7} - 3 \end{aligned}$$

The example program is written in a general form that will enable the definite integral of a cubic spline having an arbitrary number of knots to be computed. Any number of data sets may be supplied. The only changes required to the program relate to the dimensions of the arrays LAMDA and C.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      E02BDF Example Program Text.
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NC7MAX
      PARAMETER        (NC7MAX=100)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*      .. Local Scalars ..
      real            DEFINT
      INTEGER          IFAIL, J, NCAP
*      .. Local Arrays ..
      real            C(NC7MAX), K(NC7MAX)
*      .. External Subroutines ..
      EXTERNAL         E02BDF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'E02BDF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
20    READ (NIN,*) NCAP
      IF (NCAP.GT.0 .AND. NCAP+7.LE.NC7MAX) THEN
          READ (NIN,*) (K(J),J=1,NCAP+7)
          READ (NIN,*) (C(J),J=1,NCAP+3)
          IFAIL = 0
*
          CALL E02BDF(NCAP+7,K,C,DEFINT,IFAIL)
*
          WRITE (NOUT,*)
          WRITE (NOUT,99999) 'Definite integral = ', DEFINT
          GO TO 20
      END IF
      STOP
*
99999 FORMAT (1X,A,e11.3)
      END
```

9.2 Program Data

```
E02BDF Example Program Data
  7
  0.0    0.0    0.0    0.0    1.0    3.0    3.0    3.0
  4.0    4.0    6.0    6.0    6.0    6.0
 10.0   12.0   13.0   15.0   22.0   26.0   24.0   18.0
 14.0   12.0
  0
```

9.3 Program Results

```
E02BDF Example Program Results

Definite integral =  0.100E+03
```